Обратим внимание, что для сохранения ориентации контура в операторе `mesh` следует записывать внутренний контур в следующем формате `Gamma12(-5*n)` (подробнее см. гл. 17). Соответствующая строка кода, генерирующая сетку, имеет вид

```
mesh Th = buildmesh(Gamma2(5*n)+Gamma11(5*n)+Gamma3(5*n)+Gamma12(-5*n));
```

Еще один пример сложной области с отверстиями, для которой можно решать задачи (в частности, уравнение теплопроводности с заданными краевыми условиями), показан на рис. 3.7.



Рис. 3.7. Изолинии температуры в области сложной формы и триангуляция

3.3 Физические задачи, приводящие к уравнению Лапласа

Этот раздел носит справочный характер и предназначен в первую очередь для тех, кто, написав первую программу на языке FreeFem++, хочет расширить свой физический кругозор и исследовать некоторые модели реальных прикладных задач. Более полные сведения о постановке задач, конечно же, можно найти практически в любом курсе уравнений математической физики. Однако, на наш взгляд, полезно иметь возможность почерпнуть некоторую информацию непосредственно в книге о языке FreeFem++. Это, в частности, позволит сразу же провести вычислительные эксперименты и получить наглядное представление о многих физических процессах.

3.3.1 Теплопроводность

Стационарное распределение температуры в некоторой области D описывается уравнением (уравнение баланса энергии)

$$\operatorname{div} \boldsymbol{q} = f, \tag{3.15}$$

где $\boldsymbol{q}$ — плотность потока тепла, f — плотность внутренних источников.

Плотность потока тепла $\boldsymbol{q}$ связана с температурой θ законом теплопроводности Фурье

$$\boldsymbol{q} = -\varkappa \nabla \theta, \tag{3.16}$$

где $\varkappa$ — коэффициент теплопроводности ($\varkappa \geqslant 0$).

Умножая (3.15) на v и интегрируя по области D, получим

$$-\iint\limits_D \boldsymbol{q} \cdot \nabla v \, dx \, dy + \int\limits_{\partial D} v\boldsymbol{q} \cdot \boldsymbol{n} \, ds - \iint\limits_D vf \, dx \, dy = 0 \qquad (3.17)$$

или с учетом (3.16)

$$\iint\limits_D \varkappa \nabla\theta \cdot \nabla v \, dx \, dy + \int\limits_{\partial D} v\boldsymbol{q} \cdot \boldsymbol{n} \, ds - \iint\limits_D vf \, dx \, dy = 0. \qquad (3.18)$$

Далее считаем, что граница области D состоит из трех, возможно несвязных, фрагментов (некоторые Γ_i могут быть пустыми множествами)

$$\partial D = \Gamma_1 \cup \Gamma_2 \cup \Gamma_3.$$

Краевые условия первого рода, соответствующие заданию температуры на границе области, имеют вид

$$\theta\big|_{\Gamma_1} = \theta_0(x, y)\big|_{\Gamma_1}, \qquad (3.19)$$

где $\theta_0(x, y)$ — заданное распределение температуры на границе области.

С физической точки зрения такое условие соответствует случаю, когда область D «окружена» некоторой «внешней» областью («внешняя» область может быть отверстием в области D, см., например, рис. 3.6, 3.7), в которой известно распределение температуры $\theta_0(x, y)$. Кроме этого предполагается, что распределение температуры θ в области D не влияет на распределение температуры $\theta_0(x, y)$. Такое возможно, по крайней мере, по двум причинам — размеры внешней области много больше размеров области D, теплоемкость (способность тела сохранять тепло) внешней области много больше теплоемкости области D.

Краевые условия второго рода, соответствующие заданию плотности потока тепла через границу области, имеют вид

$$(\boldsymbol{q} \cdot \boldsymbol{n})\big|_{\Gamma_2} = q_n^0(x, y)\big|_{\Gamma_2}, \qquad (3.20)$$

где $q_n^0(x, y)$ — заданная плотность потока тепла через границу (точнее, нормальная к границе компонента потока тепла).

Краевые условия третьего рода имеют вид

$$(\boldsymbol{q} \cdot \boldsymbol{n})\big|_{\Gamma_3} = \varkappa_{\text{out}}(x, y)(\theta - \theta_{\text{out}}(x, y))\big|_{\Gamma_3}, \qquad (3.21)$$

где $\varkappa_{\text{out}}$ — коэффициент внешней теплопроводности ($\varkappa_{\text{out}} \geqslant 0$), $\theta_{\text{out}}(x, y)$ — температура на внешней границе области.

Случаи, когда $q_n^0 = 0$ или $\varkappa_{\text{out}} = 0$, соответствуют ***теплоизолированным*** участкам границы.

С физической точки зрения, краевые условия третьего рода (3.21) отвечают равенству нормальных компонент плотности внутреннего ($\boldsymbol{q}$) и внешнего ($\boldsymbol{q}^{\text{out}}$) потоков тепла на границе

$$(\boldsymbol{q} \cdot \boldsymbol{n})\big|_{\Gamma_3} = (\boldsymbol{q}^{\text{out}} \cdot \boldsymbol{n})\big|_{\Gamma_3}. \qquad (3.22)$$

При теплообмене по закону Ньютона считается, что нормальная компонента плотности внешнего потока тепла определяется разностью температур на внутренней (θ) и внешней (θ_{out}) границах области при помощи соотношения

$$\boldsymbol{q}^{\text{out}} = -\varkappa^{\text{out}}(\theta_{\text{out}} - \theta)\boldsymbol{n}. \qquad (3.23)$$

Легко проверить, что это соответствует потоку тепла от горячего тела к холодному ($\boldsymbol{n}$ — внешняя нормаль).

Краевое условие (3.21), соответствующее теплообмену тела с внешней средой по закону Ньютона получится при помощи (3.23) в (3.22).

Коэффициенты теплопроводности могут быть функциями, зависящими от координат. Случай $\varkappa = \varkappa(x, y)$ соответствует неоднородной области, например, состоящей из различных материалов (медь, сталь). Случай $\varkappa_{\text{out}} = \varkappa_{\text{out}}(x, y)$ моделирует неоднородности теплопроводящей границы.

Сильной формулировке задачи соответствуют уравнения (3.15), (3.16) с граничными условиями (3.19)–(3.21). Задача в слабой формулировке получится, если в (3.18) исключить при помощи условий (3.20) и (3.21) величину $(\boldsymbol{q} \cdot \boldsymbol{n})$, т. е. заменить $(\boldsymbol{q} \cdot \boldsymbol{n})$ на правые части условий (3.20) и (3.21). Для выполнения краевых условий (3.19) следует требовать выполнения на границе Γ_1 дополнительного ограничения на тестовые функции v — выполнения *однородных* условий $v|_{\Gamma_1} = 0$.

Одним из достоинств метода конечных элементов и языка FreeFem++ является возможность несложного модифицирования кода программы для решения задачи с коэффициентами, зависящими от координат.

Пусть, например, $\varkappa = \varkappa(x, y)$. Оператор, задающий функцию $\varkappa$, имеет вид (для определенности, $\varkappa(x, y) = xy$)

```
Vh kappa = x * y;
```

Эту строку следует вставить, например, между строками 22, 23 (см. с. 43).

Для решения с помощью языка FreeFem++ достаточно вместо строки 24 (см. с. 43)

```
solve Poisson(u,v) = int2d(Th)( dx(u)*dx(v) + dy(u)*dy(v) )
```

записать строку

```
solve Poisson(u,v) =
          int2d(Th)( kappa*(dx(u)*dx(v) + dy(u)*dy(v)) )
```

Особое внимание следует уделять решению второй краевой задачи Неймана, когда ***на всей границе*** поставлено краевое условие второго рода

$$(\boldsymbol{q} \cdot \boldsymbol{n})\big|_{\partial D} = q_n^0(x, y)\big|_{\partial D}. \qquad (3.24)$$

В этом случае должно выполняться так называемое условие разрешимости, которое получается интегрированием по области D уравнения (3.15) с последующей заменой $(\boldsymbol{q} \cdot \boldsymbol{n})$ при помощи (3.24)

$$\iint\limits_D \operatorname{div} \boldsymbol{q}\, dx\, dy = \iint\limits_D f\, dx\, dy \quad \Rightarrow \quad \int\limits_{\partial D} (\boldsymbol{q} \cdot \boldsymbol{n})\, ds = \iint\limits_D f\, dx\, dy$$

и окончательно

$$\int\limits_{\partial D} q_n^0\, ds = \iint\limits_D f\, dx\, dy. \tag{3.25}$$

Полученное соотношение означает, что при постановке задачи Неймана нельзя произвольно задавать плотность распределения источников тепла $f(x,y)$ и плотность потока тепла на границе $q_n^0|_{\partial D}$ — они должны удовлетворять соотношению (3.25).

Кроме этого, непосредственной подстановкой в (3.15), (3.16), (3.24) можно проверить, что если θ является решением задачи, то и $\theta + \mathrm{const}$ также будет решением задачи. Это означает, что задача Неймана имеет не единственное решение (определяется с точностью до константы).

При численном решении задачи Неймана возникают дополнительные осложнения, связанные с тем, что, по причине наличия вычислительной погрешности, точно удовлетворить условиям разрешимости (3.25) невозможно. В дальнейшем (см. гл. 8 и формулы (8.25)–(8.29)) будут показаны некоторые варианты численного решения задачи Неймана.

В заключение, укажем способ, который позволяет визуализировать векторное поле, соответствующее потоку тепла. Для этого в код программы на с. 43 следует добавить строки

```
Vh qx=-dx(u), qy=-dy(u);
plot(u, [qx,qy], coef=0.05, value=1);
```

Параметр `coef` регулирует длину стрелок векторов векторного поля при визуализации.

3.3.2 Диффузия

Предположим, что некоторая область D заполнена раствором, состоящим из растворителя и примеси. Стационарное распределение концентрации примеси описывается уравнением (уравнение баланса массы)

$$\operatorname{div} \boldsymbol{i} = r, \tag{3.26}$$

где $\boldsymbol{i}$ — плотность потока концентрации примеси, r — плотность внутренних источников концентрации, возникающих, например, в результате химических реакций.

Плотность потока концентрации $\boldsymbol{i}$ связана с концентрацией $c(x,y)$ законом Фика

$$\boldsymbol{i} = -D_c \nabla c, \tag{3.27}$$

где D_c — коэффициент диффузии ($D_c \geqslant 0$).

Далее считаем, что граница области D состоит из трех, возможно несвязных, фрагментов (некоторые Γ_i могут быть пустыми множествами)

$$\partial D = \Gamma_1 \cup \Gamma_2 \cup \Gamma_3.$$

Краевые условия первого рода, соответствующие заданию концентрации на границе области, имеют вид

$$c\big|_{\Gamma_1} = c_0(x,y)\big|_{\Gamma_1}, \tag{3.28}$$

где $c_0(x, y)$ — заданное распределение концентрации на границе области.

С физической точки зрения такое условие соответствует случаю, когда область D «окружена» некоторой «внешней» областью, в которой известна концентрация примеси $c_0(x, y)$, и предполагается, что распределение концентрации c в области D не влияет на распределение $c_0(x, y)$.

Краевые условия второго рода, соответствующие заданию плотности потока примеси через границу области, имеют вид

$$(\boldsymbol{i} \cdot \boldsymbol{n})\big|_{\Gamma_2} = i_n^0(x, y)\big|_{\Gamma_2}, \tag{3.29}$$

где $i_n^0(x, y)$ — заданная плотность потока концентрации через границу (точнее, её нормальная к границе компонента).

Краевые условия третьего рода имеют вид (ср. с (3.21)-(3.23))

$$(\boldsymbol{i} \cdot \boldsymbol{n})\big|_{\Gamma_3} = D_{\text{out}}(x, y)(c - c_{\text{out}}(x, y))\big|_{\Gamma_3}, \tag{3.30}$$

где D_{out} — коэффициент внешней диффузии ($D_{\text{out}} \geqslant 0$), $c_{\text{out}}(x, y)$ — концентрация на внешней границе области.

Случаи, когда $i_n^0 = 0$ или $D_{\text{out}} = 0$, соответствуют участкам границы, ***непроницаемым*** для концентрации примеси.

Совершенно очевидно, что с математической точки зрения процессы теплопроводности и диффузии описываются одними и теми же уравнениями. Соотношения п. 3.3.1 и п. 3.3.2 становятся одинаковыми после формальных замен

$$\theta \leftrightarrows c, \quad \boldsymbol{q} \leftrightarrows \boldsymbol{i}, \quad \varkappa \leftrightarrows D_c, \quad \varkappa_{\text{out}} \leftrightarrows D_{\text{out}},$$

$$\theta_{\text{out}} \leftrightarrows c_{\text{out}}, \quad q_n^0 \leftrightarrows i_n^0, \quad \theta_0 \leftrightarrows c_0, \quad f \leftrightarrows r.$$

3.3.3 Электрический потенциал (электростатика)

Предположим, что в диэлектрике, занимающем некоторую область D, известна плотность распределения электрического заряда $\rho_e(x, y)$. Напряженность электрического поля $\boldsymbol{E}$ связана с $\rho_e(x, y)$ одним из уравнений Максвелла

$$\operatorname{div}(\varepsilon \boldsymbol{E}) = \rho_e, \tag{3.31}$$

где ε — диэлектрическая проницаемость области D ($\varepsilon > 0$).

Электрическое поле называется потенциальным, если напряженность электрического поля может быть представлена в виде

$$\boldsymbol{E} = -\nabla \varphi, \tag{3.32}$$

где φ — потенциал электрического поля.

Для уравнений (3.31), (3.32) на фрагментах границы Γ_1 можно задавать краевые условия Дирихле

$$\varphi\big|_{\Gamma_1} = \varphi_0(x, y)\big|_{\Gamma_1}, \tag{3.33}$$

где $\varphi_0(x, y)$ — заданное распределение электрического потенциала на Γ_1.

Для остальной части границы $\Gamma_2 = \partial D \setminus \Gamma_1$ можно задавать краевые условия Неймана

$$(\boldsymbol{n} \cdot \nabla\varphi)\Big|_{\Gamma_2} = \frac{\partial\varphi}{\partial n}\Big|_{\Gamma_2} = \sigma_e(x, y)\Big|_{\Gamma_2}, \tag{3.34}$$

где $\sigma_e(x, y)$ — заданное распределение поверхностного заряда на Γ_2 (если диэлектрик граничит с проводником, то $\sigma_e(x, y) = 0$).

Формально можно рассматривать и третью краевую задачу, считая, что плотность поверхностного заряда линейно зависит от потенциала на границе. Однако физическая трактовка такого условия весьма затруднительна.

3.3.4 Электрический потенциал (проводимость)

Пусть $\boldsymbol{j}$ — плотность электрического тока, протекающего через некоторую область D. Уравнение неразрывности электрического тока (закон сохранения заряда) имеет вид

$$\operatorname{div} \boldsymbol{j} = 0. \tag{3.35}$$

Плотность электрического тока связана с напряженностью электрического поля $\boldsymbol{E}$ законом Ома

$$\boldsymbol{j} = \sigma \boldsymbol{E}, \tag{3.36}$$

где σ — электрическая проводимость области D ($\sigma \geqslant 0$).

В случае потенциального электрического поля имеем

$$\boldsymbol{j} = -\sigma\nabla\varphi, \quad \boldsymbol{E} = -\nabla\varphi. \tag{3.37}$$

Для уравнений (3.35)–(3.37) с физической точки зрения разумно ставить лишь первые и вторые краевые условия

Далее считаем, что граница области D состоит из двух, возможно несвязных, фрагментов (некоторые Γ_i могут быть пустыми множествами)

$$\partial D = \Gamma_1 \cup \Gamma_2.$$

Краевые условия первого рода, соответствующие заданию потенциала на границе области, имеют вид

$$\varphi\Big|_{\Gamma_1} = \varphi_0(x, y)\Big|_{\Gamma_1}, \tag{3.38}$$

где $\varphi_0(x, y)$ — заданное распределение электрического потенциала на Γ_1.

Заметим, что физически реализовать распределение потенциала, зависящее от координат, на каком-либо участке границы проводящей области довольно трудно. Чаще всего предполагается, что граница Γ_1 состоит из нескольких несвязных фрагментов

$$\Gamma_1 = \Gamma_{11} \cup \Gamma_{12} \ldots \Gamma_{1m}, \quad \Gamma_{1i} \cap \Gamma_{1j} = \varnothing, \quad i \neq j.$$

На каждом фрагменте Γ_{1k} задается ***постоянное*** значение потенциала

$$\varphi\Big|_{\Gamma_{1k}} = \varphi_k = \text{const}, \quad k = 1, 2, \ldots, m. \tag{3.39}$$

Краевые условия второго рода, соответствующие заданию плотности электрического тока, протекающего через границу области, имеют вид

$$(\boldsymbol{j} \cdot \boldsymbol{n})\big|_{\Gamma_2} = j_n^0(x, y)\big|_{\Gamma_2}, \tag{3.40}$$

где $j_n^0(x, y)$ — заданная плотность электрического тока через границу.

Случай $j_n^0 = 0$ соответствует электрически ***изолированным*** участкам границы. Также как и для краевых условий первого рода, чаще всего считается, что граница Γ_2 состоит из нескольких несвязных фрагментов.

3.3.5 Потенциальное течение несжимаемой жидкости

Течение жидкости называется потенциальным, если скорость течения $\boldsymbol{v}$ можно представить в виде

$$\boldsymbol{v} = \nabla\varphi. \tag{3.41}$$

Уравнение неразрывности для несжимаемой жидкости имеет вид

$$\operatorname{div} \boldsymbol{v} = 0. \tag{3.42}$$

Подставляя (3.41) в (3.42), получим

$$\Delta\varphi = 0. \tag{3.43}$$

Уравнение Бернулли, связывающее давление p со скоростью течения $\boldsymbol{v}$, записывается в форме (см. [16])

$$p + \frac{1}{2}\rho\boldsymbol{v}^2 = \text{const}, \tag{3.44}$$

где ρ — плотность жидкости ($\rho = \text{const}$).

Это соотношение, справедливое для стационарного течения невязкой жидкости, легко выводится из стационарных уравнений Эйлера

$$\rho\boldsymbol{v} \cdot \nabla\boldsymbol{v} = -\nabla p \tag{3.45}$$

или в декартовых координатах (подразумевается суммирование по повторяющимся индексам)

$$\rho v_k \frac{\partial v_i}{\partial x_k} = -\frac{\partial p}{\partial x_i}. \tag{3.46}$$

Подставляя (3.41) в (3.46), получим

$$\rho\frac{\partial\varphi}{\partial x_k}\frac{\partial}{\partial x_k}\frac{\partial\varphi}{\partial x_i} = -\frac{\partial p}{\partial x_i}, \quad \rho\frac{\partial\varphi}{\partial x_k}\frac{\partial}{\partial x_i}\frac{\partial\varphi}{\partial x_k} = -\frac{\partial p}{\partial x_i}, \quad \frac{1}{2}\rho\frac{\partial}{\partial x_i}\left(\frac{\partial\varphi}{\partial x_k}\right)^2 = -\frac{\partial p}{\partial x_i}.$$

Интегрируя, выводим (3.44)

$$\frac{1}{2}\rho\frac{\partial}{\partial x_i}\left(\frac{\partial\varphi}{\partial x_k}\right)^2 = -\frac{\partial p}{\partial x_i}, \quad \frac{1}{2}\rho\left(\frac{\partial\varphi}{\partial x_k}\right)^2 + p = \text{const}, \quad \frac{1}{2}\rho\boldsymbol{v}^2 + p = \text{const}.$$

В двумерном случае уравнению неразрывности можно удовлетворить, вводя функцию тока $\psi(x, y)$

$$u = \psi_y, \quad w = -\psi_x, \quad \boldsymbol{v} = (u, w), \tag{3.47}$$

где u, w — соответственно, компоненты скорости вдоль осей x и y.

Действительно, при дополнительных требованиях на гладкость (для возможности изменения порядка дифференцирования) имеем

$$\operatorname{div} \boldsymbol{v} = u_x + w_y = \psi_{yx} - \psi_{xy} \equiv 0.$$

Для функции тока ψ в случае потенциального течения справедливо уравнение Лапласа

$$\Delta\psi = 0, \quad (u_y - w_x = \psi_{yy} + \psi_{xx}, \quad u_y - w_x = \varphi_{xy} - \varphi_{yx} = 0). \tag{3.48}$$

Для задач гидродинамики довольно затруднительно перечислить типичные виды краевых условий. Эти условия могут быть и первого, и второго, и третьего рода. Ограничимся лишь рассмотрением одной из задач, возникающих в области аэродинамики, напомнив, что рассматриваемые уравнения справедливы не только для жидкостей, но и для не очень разреженных газов, таких, например, как воздух.

3.3.5.1 Стационарное обтекание крыла

Предположим, что в неограниченной области имеется некоторый контур W (wing), моделирующий крыло (см. рис. 3.8), и обтекаемый потоком газа (воздуха). Считаем, что контур непроницаем для газа и вдали от контура (на бесконечности) скорость течения постоянна и направлена вдоль некоторого вектора

$$\boldsymbol{v}\big|_\infty = \boldsymbol{v}_\infty, \quad \boldsymbol{v}_\infty = (u_\infty, w_\infty), \quad \operatorname{tg}\alpha = \frac{w_\infty}{u_\infty}, \tag{3.49}$$

где u_∞, w_∞ — заданы и α — угол наклона направления течения к оси x.

Воспользовавшись определением (3.47), легко убедиться, что функция тока $\psi(x, y)$ на бесконечности задается соотношением (с точностью до несущественной константы)

$$\psi\big|_\infty = \psi_\infty = u_\infty y - w_\infty x. \tag{3.50}$$

Для определения ψ имеем следующую задачу

$$\Delta\psi = 0, \quad \psi\big|_W = 0, \quad \psi\big|_\infty = \psi_\infty. \tag{3.51}$$

Краевое условие $\psi|_W = 0$ на контуре W отвечает случаю контура, непроницаемого для газа.

Решив задачу (3.51) и используя (3.47), можно найти поле скоростей и при помощи уравнения Бернулли (3.44) определить давление p (естественно, с точностью до константы)

$$\boldsymbol{v} = (u, w) = (\psi_y, -\psi_x), \quad p = -\frac{1}{2}\boldsymbol{v}^2 = -\frac{1}{2}\left(\psi_x^2 + \psi_y^2\right). \tag{3.52}$$

Ясно, что невозможно численно решать задачу (3.51) в бесконечной области. Поэтому условие $\psi|_\infty = \psi_\infty$ заменим условием на некоторой окружности достаточно большого радиуса R

$$\psi\Big|_{(x^2+y^2)=R^2} = \psi_\infty.$$

Приведем код программы на языке FreeFem++, задавая, для определенности, в качестве контура W так называемый профиль крыла NACA2412 (способ конструирования различных профилей приведен на с. 57) и выбирая следующие параметры

$$R = 5, \quad u_\infty = 1{,}0, \quad w_\infty = 0{,}5.$$

```
int Sp=97, Sm=98; // метки контура
int n=1;
real R=5, R0=0.8, shift=0.5; // параметры
real p0, m0, t0; // параметры контура крыла
// NACA2412
m0 = 0.01*2;      p0 = 0.1*4;     t0 = 0.01*12;
// функция для задания толщины крыла
func real Yt(real t)
{ return  5*t0*(0.296900*sqrt(t)- 0.126556*t - 0.356307*t^2
          + 0.290672*t^3 - 0.104709*t^4);                        }
// функция для задания средней линии крыла
func real Yc(real t)
{ return (0<=t)*(t<=p0)*(m0*(1/(p0^2))*(2*p0*t -t^2))+
         (p0<=t)*(t<=1.0)*(m0*((1/(1-p0)^2)*((1-2*p0)+2*p0*t-t^2))); }
//  theta=arctg(dYc/dt)
func real theta(real t)
{ return   atan((0<=t)*(t<=p0)*(m0*(1/(p0^2))*(2*p0 -2*t))+
           (p0<=t)*(t<=1.0)*(m0*((1/(1-p0)^2)*(2*p0-2*t))));    }
// задание профиля крыла
border C(t=0,2*pi) { x=5*cos(t); y=5*sin(t);}
border WingUp(t=0,1){x = t-Yt(t)*sin(theta(t));
                     y = Yc(t)+ Yt(t)*cos(theta(t)); label=Sp; }
border WingDn(t=1,0){x = t+Yt(t)*sin(theta(t));
                     y = Yc(t)- Yt(t)*cos(theta(t)); label=Sm; }
mesh Th= buildmesh(C(50*n)+WingUp(70*n)+WingDn(70*n));
// задание  пространства конечных элементов
fespace Vh(Th,P2); Vh psi,v;
// вспомогательный контур для визуализации в окрестности крыла
border c0(t=0,2*pi) { x=R0*cos(t)+shift; y=R0*sin(t); }
mesh Zoom = buildmesh(c0(30*n)+WingUp(35*n)+WingDn(35*n));
plot(c0(50*n)+WingUp(70*n)+WingDn(70*n),wait=1);
fespace ZVh(Zoom,P2);
ZVh Zpsi,Zcp,Zu,Zw;
// направление потока воздуха
real Uinfty = 1.0,  Winfty = 0.5;
solve potential(psi,v) = int2d(Th)(dx(psi)*dx(v)+dy(psi)*dy(v))
                         + on(C, psi = Uinfty*y-Winfty*x)
                         + on(Sp, Sm, psi=0);
```

```
// вычисление давления и компонент скорости
Vh u = dy(psi), w = -dx(psi), cp = -u^2-w^2;
// вспомогательные операторы для визуализации
Zpsi = psi;   Zu = u;   Zw = w;   Zcp = cp;
plot(Zpsi, wait=1, bw=1, nbiso=50);
plot(Zcp, wait=1, bw=1, nbiso=40);
plot([Zu,Zw], wait=1, bw=1, coef=0.02);
```

Результаты расчетов представлены на рис. 3.8.



Рис. 3.8. Изолинии функции тока, давления и поле скоростей при обтекании крыла

Коротко опишем один из способов конструирования профилей крыла, известных как семейство профилей `NACA 4 digits`[1].

Задается стандартная функция $y_t(t)$, называемая толщиной крыла, и функция $y_c(t)$, называемая средней линией контура крыла (chamber line or mean line)

$$y_t(t) = 5T(0{,}296900\sqrt{t} - 0{,}126556\,t - 0{,}356307\,t^2 + 0{,}290672\,t^3 - 0{,}104709\,t^4),$$

$$y_c(t) = \begin{cases} \dfrac{m}{p^2}(2pt - t^2), & 0 \leqslant t \leqslant p, \\ \dfrac{m}{(1-p^2)}(1 - 2p + 2pt - t^2), & p \leqslant t \leqslant 1. \end{cases}$$

Вводится вспомогательная функция $\theta(t)$

$$\theta(t) = \operatorname{arctg}\left(\frac{dy_c(t)}{dt}\right), \quad \frac{dy_c(t)}{dt} = \begin{cases} \dfrac{2m}{p^2}(p - t), & 0 \leqslant t \leqslant p, \\ \dfrac{2m}{(1-p^2)}(p - t), & p \leqslant t \leqslant 1. \end{cases}$$

Координаты верхней (upper) и нижней (lower) частей контура крыла определяются параметрическими уравнениями

$$x_{\text{upper}} = t - y_t(t)\sin(\theta(t)), \quad y_{\text{upper}} = y_c(t) + y_t(t)\cos(\theta(t)), \quad 0 \leqslant t \leqslant 1,$$
$$x_{\text{lower}} = t + y_t(t)\sin(\theta(t)), \quad y_{\text{lower}} = y_c(t) - y_t(t)\cos(\theta(t)), \quad 0 \leqslant t \leqslant 1.$$

[1] См., например, `http://www-berkeley.ansys.com/cfd/naca.html`

Параметры T, m, p, входящие в указанные соотношения, вычисляются по цифровому коду профиля. Например, для профиля крыла `NACA2412`, используя цифры кода в качестве числителей дробей, получим (см. рис. 3.9)

$$m = \frac{2}{100}, \quad p = \frac{4}{10}, \quad T = \frac{12}{100}.$$



Рис. 3.9. Профиль крыла NACA2412 (справа в масштабе 1:1)

На рис. 3.10 для сравнения показаны некоторые профили крыла `NACA`.



Рис. 3.10. Различные профили крыла NACA